RI 8926

Bureau of Mines Report of Investigations/1985

Mathematical Simulation of Automated Metabolic Breathing Simulator and Self-Contained Self-Rescuer

By John C. Edwards

UNITED STATES DEPARTMENT OF THE INTERIOR

Report of Investigations 8926

Mathematical Simulation of Automated Metabolic Breathing Simulator and Self-Contained Self-Rescuer

By John C. Edwards

UNITED STATES DEPARTMENT OF THE INTERIOR
Donald Paul Hodel, Secretary

BUREAU OF MINES
Robert C. Horton, Director

Library of Congress Cataloging in Publication Data:

Edwards, John C

Mathematical simulation of automated metabolic breathing simulator and self-contained self-rescuer.

(Report of investigations ; 8926)

Includes bibliographic references.

Supt. of Docs. no.: I 28.23:8926.

1. Respiration—Mathematical models. 2. Respirators—Mathematical models. 3. Metabolism—Mathematical models. 4. Mine rescue work—Equipment and supplies. I. Title. II. Series: Report of investigations (United States. Bureau of Mines) ; 8926.

TN23.U43 [QP121] 622s [620.8'6] 84-600268

CONTENTS

ILLUSTRATIONS

TABLES

UNIT OF MEASURE ABBREVIATIONS USED IN THIS REPORT

b/min	breath per minute	min	minute
L	liter	mph	mile per hour
L/min	liter per minute	pct	percent
L/s	liter per second	s	second

MATHEMATICAL SIMULATION OF AUTOMATED METABOLIC BREATHING SIMULATOR AND SELF-CONTAINED SELF-RESCUER

By John C. Edwards [1]

ABSTRACT

The Bureau of Mines developed a mathematical model that decribes the interaction of an automated breathing and metabolic simulator, or lung, with a self-contained self-rescuer that is supplied compressed oxygen. The model can be used to predict the gas partial volumes of O_2, N_2, and CO_2 in the lung and in the self-rescuer breathing bag for a prescribed metabolic state that characterizes the lung. Nonlinear rate equations are used to describe the temporal evolution of the gas partial volumes. A numerical procedure was established with a FORTRAN computer program to solve these equations, which correspond to a given metabolic state. Examples are presented of applications of the model to predict bag volume and bag oxygen concentration changes with time when the bag is controlled by an automated breathing simulator. Under certain restrictive conditions, the model equations can also be used to analytically predict the time required for the bag to reach a maximum or minimum volume. The model equation in this latter case was extended to transitions between metabolic states, with transitions described as exponential growth or decay processes.

[1] Physicist, Pittsburgh Research Center, Bureau of Mines, Pittsburgh PA.

INTRODUCTION

As part of a continuing research program in the area of postdisaster survival and rescue, the Bureau of Mines has sponsored research to provide the optimum self-contained self-rescuer (SCSR) for in-mine use. An optimum SCSR would satisfy design requirements based upon the National Academy of Engineering (NAE) committee's recommendations.[2] This research program has supported the development of SCSR's that store oxygen either chemically or as a compressed gas. The results of the program have been described in detail,[3] as well as the laboratory testing of SCSR's for durability and reliability.[4] Testing of SCSR's has involved both human subjects and breathing simulators. A breathing metabolic simulator, as described by Kyriazi,[5] has associated with it physiological parameters corresponding to a metabolic state, such as oxygen consumption, CO_2 production, breathing rate, and tidal volume. Each metabolic state may well represent the physiological condition of a miner performing certain tasks. It is important that the SCSR supply the subject with adequate oxygen and absorb efficiently the CO_2 produced. As part of the testing program, it is useful to have an analytic procedure to evaluate the response of SCSR's to various metabolic states. For this purpose a mathematical model was developed that predicts the breathing bag volume and gas concentration under various SCSR operating conditions.

MODEL DESCRIPTION

A schematic of the total system, the automated metabolic breathing simulator (AMBS) (lung) and SCSR, is shown in figure 1. The lung volume U, which is the sum of the partial volumes U(k) for each gas species k, changes according to a prescribed function of time corresponding to inhalation and exhalation with a maximum lung volume change equal to the tidal volume V_T. The bag volume V, which is similarly the sum of the partial volumes V(k), responds to volume changes in the lung. The gas, assumed incompressible, has three components: O_2, N_2, and CO_2. Oxygen is irreversibly lost from the lung during inhalation at the rate $\dot{V}_{O_2}$, referred to as oxygen uptake, and is reversibly exchanged through the mouthpiece with the breathing bag during inhalation and exhalation. CO_2 enters the lung from the alveoli at a production rate $\dot{V}_{CO_2}$ during exhalation and is reversibly exchanged through the mouthpiece with the breathing bag during the breathing cycle. The ratio of $\dot{V}_{CO_2}$ to $\dot{V}_{O_2}$ is defined to be the respiratory quotient, RQ. The CO_2 is absorbed, generally with near 100 pct efficiency, within the bag in this model. The bag volume responds to volume changes in the lung, while receiving a constant oxygen supply from the oxygen bottle. Supplemental oxygen is supplied to the bag if it collapses sufficiently to trigger a demand valve. This additional oxygen supply mode is modeled by assuming the demand valve opens whenever the bag volume is less than an experimentally predetermined value. When the bag volume reaches a maximum value, a relief valve

[2]Committee on Mine Rescue and Survival Techniques, National Academy of Engineering. Mine Rescue and Survival. Final Report (contract SO190616). BuMines OFR 4-70, 1970, 81 pp.; NTIS PB 191 691.

[3]Kovac, J. G. An Overview of Oxygen Self-Rescuer Technology. Paper in Postdisaster Survival and Rescue Research. Proceedings: Bureau of Mines Technology Transfer Seminar, Pittsburgh, Pa., November 16, 1982. BuMines IC 8907, 1982, pp. 3-17.

[4]Kyriazi, N. Laboratory Environmental Testing of Chemical Oxygen Self-Rescuers for Ruggedness and Reliability. Paper in Postdisaster Survival and Rescue Research. Proceedings: Bureau of Mines Technology Transfer Seminar, Pittsburgh, Pa., November 16, 1982. BuMines IC 8907, 1982, pp. 18-31.

[5]Work cited in footnote 4.



FIGURE 1. - Schematic of AMBS-SCSR system.

opens and gas escapes until the bag volume starts to decrease.

The lung-SCSR is not a closed system. Leakage can occur around the mouthpiece during both inhalation and exhalation. Leakage is included in the model as an empirically determined quantity.

The change in the partial volume of the individual gas species can be simulated by rate equations. The driving force for the rate equations is the lung expansion and contraction representative of a given metabolic state. Separate control volumes are displayed by the dashed lines in figure 1 for the lung and bag. The gas partial volumes, U(k) and V(k), are indexed according to gas species O_2, and CO_2, and N_2, respectively. The rate equations are constructed by systematically analyzing the influx of gases across each control volume in figure 1. Expiration of gases from the lung results in a net volumetric flow rate $\dot{Q}E(k) - \dot{Q}L_e(k)$ of gas species k from the lung into the bag, where $\dot{Q}E(k)$ is due to expiration and $\dot{Q}L_e(k)$ is the leakage rate of species k from the breathing apparatus during expiration. Inspiration of gases into the lung results in a net flow $\dot{Q}I(k) + \dot{Q}L_i(k)$ of gas species k from the bag into the lung, where $\dot{Q}I(k)$ is the volumetric flow rate due to inspiration and $\dot{Q}L_i(k)$ is the leakage of species k into the breathing apparatus during inspiration. The quantities $\dot{Q}I(k)$ and $\dot{Q}E(k)$ are proportional to the total inspiration and expiration volume rates, $\dot{Q}I$ and $\dot{Q}E$, and the species concentrations, as follows:

$$\dot{Q}I(k) = \frac{V(k)}{V}\dot{Q}I; \tag{1}$$

$$\dot{Q}E(k) = \frac{U(k)}{U}\dot{Q}E. \tag{2}$$

In addition to leakage, the bag vents gas to the external environment whenever it reaches a maximum permissible volume, V_m. When this occurs, a relief valve opens and each species k is released at a rate $\dot{Q}_x V(k)/V_m$. The flow rate $\dot{Q}_x$ is simply $-\frac{dV}{dt}$.

Oxygen is supplied to the bag from the oxygen cylinder through two modes. The first, and primary, is a constant flow rate $\dot{Q}B$; the second is activated only if the bag volume collapses below a threshold volume and triggers a demand valve, in which case an additional oxygen flow rate, $\dot{Q}D$, occurs.

CO_2 is removed from the system through absorption by LiOH as it passes into the bag during exhalation. The absorption rate, $\dot{Q}A$, occurs with the efficiency η related to the CO_2 production by

$$\eta = \dot{Q}A/\dot{V}_{CO_2}. \tag{3}$$

More experimental work is required to determine an explicit representation of $\dot{Q}A$ in terms of a particular breathing bag.

Based upon the assumption of uniform mixing of gases in the bag, the time rate of change of the partial volume V(k) is expressed by

$$\frac{dV(k)}{dt} = \dot{Q}E(k) - \dot{Q}L_e(k) - \dot{Q}I(k) + \dot{Q}L_i(k) + (\dot{Q}B + \dot{Q}D)\,\delta_{k,O_2} - \dot{Q}A\,\delta_{k,CO_2} - \dot{Q}_x \frac{V(k)}{V_m}, \tag{4}$$

where $\delta_{i,j}$ is the Kronecker delta function, where $\delta_{i,j} = 1$ if $i = j$, and zero otherwise.

The leakage terms for the constitutive gases are defined in terms of the specific volume fraction and total leakage rates.

$$\dot{Q}L_i(k) = \frac{V(k)}{V} \dot{Q}L_i; \tag{5}$$

$$\dot{Q}L_e(k) = \frac{U(k)}{U} \dot{Q}L_e. \tag{6}$$

The volumetric leakage rates $\dot{Q}L_i$ and $\dot{Q}L_e$ depend upon the degree to which the mouthpiece of the SCSR fits the user and must be determined experimentally.

Equation 1 through 6 are combined to rewrite equation 4 as

$$\frac{dV(k)}{dt} = \frac{U(k)}{U}(\dot{Q}E - \dot{Q}L_e) - \frac{V(k)}{V}(\dot{Q}I - \dot{Q}L_i) + (\dot{Q}B + \dot{Q}D)\,\delta_{k,O_2}, - \eta\,\dot{V}_{CO_2}\delta_{k,CO_2} - \frac{V(k)}{V_m}\dot{Q}_x. \tag{7}$$

Equation 7 constitutes a set of nonlinear first-order ordinary differential equations. A solution is determined when equation 7 is coupled with the corresponding rate equations for the gas partial volume changes in the lung, or breathing simulator.

An analysis similar to the one applied to the breathing bag is applied to the control volume about the lung. Gas exchange occurs with both the pulmonary capillary and the breathing bag. Oxygen is lost irreversibly from the lung to the pulmonary cavity as oxygen uptake, while CO_2 is produced. The governing rate equation for partial volume U(k) of species k in the lung is

$$\frac{dU(k)}{dt} = \dot{Q}I(k) - \dot{Q}E(k) - \dot{V}_{O_2}\delta_{k,O_2} + \dot{V}_{CO_2}\delta_{k,CO_2}. \tag{8}$$

Equation 8 is rewritten with the aid of equations 1 and 2 as

$$\frac{dU(k)}{dt} = \frac{V(k)}{V}\dot{Q}I - \frac{U(k)}{U}\dot{Q}E - \dot{V}_{O_2}\delta_{k,O_2} + \dot{V}_{CO_2}\delta_{k,CO_2}. \tag{9}$$

The model equations, 7 and 9, can be solved under a variety of breathing simulations to predict the adequacy of experimentally controllable parameters (such as minimum and maximum bag volume, oxygen supply, CO_2 absorption efficiency, and gas leakage at the mouthpiece) for ensuring safe operating conditions for the self-rescuer. Equations 7 and 9 represent a set of six coupled ordinary differential equations. An analytic solution to these equations is not possible under the most general conditions. For this reason, a generalized numerical procedure was established through a FORTRAN computer program, which determines a numerical solution to equations 7 and 9. The program solves a set of coupled nonlinear algebraic equations that constitute a finite difference representation, implicit in time, of the model rate equations. A listing of the program is found in appendix B.

For application of the model, it is necessary to specify the imposed volumetric change in the lung, or breathing simulator, as a function of time. This requires a specification of the amplitude and frequency of the breathing cycle. The amplitude of the oscillation is one-half of the tidal volume. For the applications discussed below, constant-frequency, constant-amplitude sinusoidal functions are considered. It is shown in the next section, "Analytic Model," how the time for the bag to reach a maximum or minimum volume may be estimated.

The inhalation and exhalation for a constant-frequency, constant-amplitude breathing process described by a sinusoidal function can be written as

$$\dot{Q}I(t) = \frac{1}{2} V_T \omega \sin(\omega t) + \dot{V}, \quad 2n\pi < \omega t < (2n+1)\pi, \quad (10)$$

$$\dot{Q}E(t) = -\frac{1}{2} V_T \omega \sin(\omega t) - \dot{V}, \quad (2n-1)\pi < \omega t < 2n\pi, \quad (11)$$

where $\dot{V} = \dot{V}_{O_2} - \dot{V}_{CO_2}$

for $n = 0, 1, 2.$

The angular frequency ω, in units of radians per second, is related to the breathing rate $\dot{R}$ by

$$\omega = 2\pi\dot{R}/60. \quad (12)$$

Two applications were made of the model to metabolic states representative of a miner performing tasks under a given set of operating conditions. Table 1 shows cases selected from Kamon[6] and analyzed with the program in appendix B.

Case S1 corresponds to a miner walking level at 3 mph, while case S2 corresponds to a miner running up a hill that has a grade of 2.5 pct, at a speed of 5.8 mph.

For both cases the initial oxygen concentration in the lung and breathing bag was 20 pct, and the initial nitrogen concentration was 80 pct. A constant tidal volume of 1.29 L was used.

TABLE 1. - Selected metabolic states

Parameters	S1	S2
Oxygen uptake ($\dot{V}_{O_2}$)...L/min..	1.29	2.91
Respiratory quotient (RQ)....	0.94	1.13
Breathing rate ($\dot{R}$)....b/min..	20.94	58.4

For the simulations made with the model for cases S1 and S2, the following conditions were assumed: the CO_2 absorption efficiency was 100 pct; the constant oxygen rate supply was 1.5 L/min; the maximum, V_m, and minimum, V_{min}, bag volumes were 5 and 1 L, respectively; and the initial bag volume was 4.5 L. The minimum time either the demand or relief valve would be open when activated was 0.25 s. Figures 2 through 5 show the bag volume and bag oxygen concentration for case S1

[6]Kamon, E., T. Bernard, and R. Stein. Steady State Respiratory Response to Tasks in Federal Testing of Self-Contained Breathing Apparatus. Am. Ind. Hyg. Assoc. J. v. 36, Dec. 1975, pp. 886-896.



FIGURE 2. - Breathing bag volume for case S1 during first minute.



FIGURE 3. - Oxygen concentration in breathing bag for case S1 during first minute.



FIGURE 4. - Breathing bag volume for case S1 during second minute.

during the first and second minutes of use. Figures 2 and 4 show that the maximum extension of the volume slowly increases. Figures 3 and 5 show that there is a rapid increase in the oxygen concentration during the initial 15 s, followed by a nearly constant rate of increase during the ensuing 90 s.

The evolution of the bag volume and bag oxygen concentration for case S2 during the initial 2 min is shown in figures 6 through 9. Figures 6 and 7 show a constant rate decrease in the volume and oxygen concentration during the first minute. Figure 8 shows that after approximately 95 s the bag reaches a volume of 1 L, V_{min}, at which time the demand valve opens and supplies additional oxygen to the bag. Figure 9 shows an increasing oxygen concentration after the demand valve opens.

These simulations show how the program can be used to predict the response of the SCSR breathing bag to various metabolic states, where each state is defined by breathing rate, respiratory quotient, and oxygen uptake.



FIGURE 5. - Oxygen concentration in breathing bag for case S1 during second minute.



FIGURE 6. - Breathing bag volume for case S2 during first minute.



FIGURE 7. - Oxygen concentration in breathing bag for case S2 during first minute.



FIGURE 8. - Breathing bag volume for case S2 during second minute.



FIGURE 9. - Oxygen concentration in breathing bag for case S2 during second minute.

ANALYTIC MODEL

The changes in the bag volume can be predicted prior to the time the bag reaches its maximum or minimum volume from an analytic model. The lung volume is assumed to vary according to a sinusoidal function of time, with time-dependent changes in tidal volume, oxygen uptake, and breathing rate permitted as a simulation of differences between metabolic states. In particular, the lung volume is assumed to be described by

$$U(t) = A - 1/2\ V_T(t) \cos(\omega(t)t), \tag{13}$$

where the breathing rate, ω, is a function of time.

The model is readily analyzed with the assumption that a transition between two metabolic states occurs according to an exponential growth or decay that defines changes in the tidal volume, breathing rate, and oxygen uptake over a time period τ.

$$\omega(t) = \omega(o)\exp(\lambda_1 t); \tag{14}$$

$$V_T(t) = V_T(o)\exp(\lambda_2 t); \tag{15}$$

$$\dot{V}_{O_2}(t) = \dot{V}_{O_2}(o)\exp(\lambda_3 t). \tag{16}$$

The rate equation for the total bag volume follows from equation 4 under the assumptions that (1) no leakage occurs, (2) CO_2 is ignored, and (3) the bag volume has not reached either a maximum or minimum value that would trigger the relief or the demand valve. Under these conditions, the rate equation for the bag volume is represented as follows:

$$\frac{dV}{dt} = \dot{Q}E - \dot{Q}I + \dot{Q}B, \tag{17}$$

where $\quad \dot{Q}E = -\dfrac{dU}{dt} - \dot{V}_{O_2}$

and $\quad \dot{Q}I = \dfrac{dU}{dt} + \dot{V}_{O_2},$

for the appropriate part of the breathing cycle as defined by equations 10 and 11.

Substitution of equations 13 through 16 into equation 17 yields a nonlinear rate equation for the bag volume:

$$\frac{dV}{dt} = 0.5\frac{d}{dt}[V_T(t)\cos(\omega(t)t)] - \dot{V}_{O_2}(t) + \dot{Q}B \tag{18}$$

$$= -0.5\ V_T(t)[\omega(t)(1 + \lambda_1 t)\sin(\omega(t)t) - \lambda_2\cos(\omega(t)t)] - \dot{V}_{O_2}(t) + \dot{Q}B. \tag{19}$$

A numerical procedure was developed to solve equation 19 and written as a FORTRAN computer program, separate from the program presented in appendix B. With suitable modifications, the program in appendix B could be extended to include time-dependent metabolic states.

The response of the bag to several metabolic states and transitions between states was simulated. The states considered are shown in table 2.

The bag volume was determined as a function of time for a transition from state 1 to state 2 over a 30-s time interval, as well as for residence in each of the five states shown in table 2. The initial, maximum, and minimum bag volumes were 4.5, 5, and 1 L, respectively. Oxygen flow into the bag was at a constant rate $\dot{Q}B$ of 1.5 L/min.

TABLE 2. - Metabolic states

State	Breathing rate ($\dot{R}$), b/min	Tidal volume (V_T), L	Oxygen uptake ($\dot{V}_{O_2}$), L/min
1	18.0	2.01	1.50
2	34.5	3.50	3.0
3	34.5	2.01	3.0
4	34.5	1.5	3.0
5	34.5	3.1	3.0

The results for the six processes discussed above are shown in table 3.

TABLE 3. - Response of breathing bag to metabolic states

Process	Bag volume response
Transition from state 1 to state 2.	Reaches V_m in 31 s.
State 1.............	Never exceeds V(o).
State 2.............	Reaches V_{min} in 0.8 s.
State 3.............	Reaches V_{min} in 62 s.
State 4.............	Reaches V_{min} in 81 s.
State 5.............	Reaches V_{min} in 18 s.

The significant departure of the bag response for a transition between states 1 and 2 from the response in either state is due to the difference in rate change of tidal volume and oxygen uptake.

Those cases in table 3 that represent residence in a single state can be estimated from equation 19 under the assumption $\lambda_1 = 0$ for i = 1,3. Equation 19 can be integrated to yield V(t).

$$V(t) = V(o) + 0.5\ V_T(o)\ (\cos(\omega(o)t) - 1)$$
$$- \dot{V}_{0_2}\ t + \dot{Q}Bt. \qquad (20)$$

State 1 represents cases where the bag volume never reaches a minimum, nor increases beyond V(o). This state represents the case in which the oxygen production is balanced by the oxygen uptake. Under this constraint, equation 20 can be written

$$V(t) = V(o) + 0.5\ V_T(o)$$
$$(\cos(\omega(o)\ t) - 1). \qquad (21)$$

The second term in equation 21 oscillates between $-V_T(o)$ and o. This implies that V(t) oscillates between $V(o) - V_T(o)$ and V(o). This shows V(t) never exceeds V(o). Since, for the values considered, $V(o) - V_T(o) > V_{min}$, the bag never reaches the minimum volume.

PROGRAM USAGE

The FORTRAN computer program, listed in appendix B, is written in modular form so that additions can be made to the program as required. Input data is entered into the program through definitions in the subroutine INIT as well as through the input data file, DUMMIE.DAT, which is read into the program in INIT. The format for DUMMIE.DAT is shown in table 4.

There are appropriate comment cards in the program that describe required input data and the organizational structure of the program.

TABLE 4. - Input data on file DUMMIE.DAT

Column	Format	Symbol	Definition of symbol
LINE 1: OXYGEN PRODUCTION, UPTAKE, AND RESPIRATORY QUOTIENT			
6-10	F5.2	BO2	Oxygen production from bottle, L/min.
16-20	F5.2	UO2	Oxygen uptake, L/min.
26-30	F5.2	RQ	Respiratory quotient.
35-42	F5.2	RATTE	Breathing rate, b/min.
LINE 2: INITIAL GAS CONCENTRATIONS IN BAG, PCT			
6-10	F5.2	BCNO2	Initial O_2.
16-20	F5.2	BCNCO2	Initial CO_2.
26-30	F5.2	BCNN2	Initial N_2.
LINE 3: INITIAL GAS CONCENTRATION IN LUNG, PCT			
6-10	F5.2	UCNO2	Initial O_2.
16-20	F5.2	UCNCO2	Initial CO_2.
26-30	F5.2	UCNN2	Initial N_2.

CONCLUSIONS

An operational mathematical model was developed that simulates the interaction of an automated metabolic breathing simulator, or lung, with a self-contained breathing apparatus. The model is available as a FORTRAN computer program. It was demonstrated how the oxygen concentration in the breathing bag and bag volume change when subjected to a sinusoidal breathing rate.

A simplified analytic model was developed that can be used to estimate the time for the breathing bag to reach a minimum or maximum volume for a given metabolic state of the lung. For transition between metabolic states, a numerical solution was developed for the bag volume. The transition between states was assumed to obey an exponential growth or decay relationship. The result for the transition between two metabolic states is significantly different from the result for residence in either metabolic state. This further illustrates the advantage of a mathematical model for exploring the response of a breathing bag under a variety of driving forces.

The computer program will be used in conjunction with an automated breathing and metabolic simulator to evaluate the capability of various self-contained self-rescuers under a variety of operating conditions corresponding to metabolic states.

APPENDIX A.--LIST OF SYMBOLS

$\frac{d}{dt}$	derivative with respect to time, s^{-1}
$\dot{Q}A$	CO_2 absorption rate, L/s
$\dot{Q}B$	constant oxygen supply from bottle to breathing bag, L/s
$\dot{Q}D$	oxygen supply on demand to breathing bag, L/s
$\dot{Q}E(k)$	volumetric flow rate of gas species k from lung into breathing bag during exhalation, L/s.
$\dot{Q}I(k)$	volumetric flow rate of gas species k from breathing bag into lung during inhalation, L/s
$\dot{Q}L_e(k)$	leakage rate of gas species k from breathing apparatus during exhalation, L/s
$\dot{Q}L_i(k)$	leakage rate of gas species k into breathing apparatus during inhalation, L/s
$\dot{R}$	breathing rate, b/min
RQ	respiratory quotient, 1
t	time, s
U(k)	gas partial volume of species k in lung, L
U(t)	lung volume at time t, L
V(k)	gas partial volume of species k in breathing bag, L
V(t)	breathing bag volume at time t, L
$\dot{V}_{CO_2}$	CO_2 production rate, L/min
V_m	maximum bag volume, L
V_{min}	minimum bag volume, L
$\dot{V}_{O_2}$	oxygen uptake, L/min
V_T	tidal volume, L
η	CO_2 absorption efficiency
δ	Kronecker delta function
ω	angular frequency, s^{-1}
π	pi, equals 3.145
τ	transition time between metabolic states, s

APPENDIX B.--LISTING OF FORTRAN COMPUTER PROGRAM

```
C     PROGRAM TO CALCULATE PARTIAL GAS VOLUMES IN LUNG AND BAG
      INCLUDE 'LUNGG.COM'
      EXTERNAL FDC,FET
      LP=6
      ILP=2
      OPEN(UNIT=ILP,NAME='GASPLOT3.DAT',STATUS='NEW')
      EPS=1.0E-06
      CALL INIT
      WRITE(LP,2000) VMAX,VMIN,VT,FREQ,B1,B2
 2000 FORMAT(4X,'VMAX=',E12.5,'CUCM',2X,'VMIN=',E12.5,'CUCM',2X,
     1'VT=',E12.5,'CUCM',2X,'FREQ=',E12.5,'RADS/SEC'/4X,'B1=',E12.5
     2,'CUCM/SEC',2X,'B2=',E12.5,'CUCM/SEC'/)
      CALL OUTPT
      TIME=0.
      MAX=70050
      MAXL=MAX-600-1
      MAX1=1000
      MAX2=1200
      MAX=600
      MAXL=MAX-600
      MAXL=0
      MOP=1
      NCT=0
C     INCREMENT TIME STEP
      ILK=0
 1000 CONTINUE
C     IF(ILK.EQ.1) MAX=NNCT+50
      NCT=NCT+1
      MCT=0
      KCLT=1
      MCLT=1
      MM=FREQ*TIME/PI
      IF(MOD(MM,2).EQ.0) P4=B1
      IF(MOD(MM,2).EQ.1) P5=B2
      FCOEF=0.
      GCOEF=0.
      IF(MOD(MM,2).EQ.0) FCOEF=1.
      IF(MOD(MM,2).EQ.1) GCOEF=1.
      IFLAG=0
      JFLAG=0
      IF(MOD(MM,2).EQ.0) IFLAG=1
      IF(MOD(MM,2).EQ.1) JFLAG=1
C     FP: LUNG INHALATION
C     GP: LUNG EXHALATION
      P4=B1-B2
      P5=-P4
      FP=AA*SIN(FREQ*TIME)+P4
      FP=FCOEF*FP
      GP=-AA*SIN(FREQ*TIME)+P5
```

```
      GP=GCOEF*GP
      FL=0.
      FL=FCOEF*FL
      GL=0.
      GL=GCOEF*GL
C     FP=ABS(FP)
C     GP=ABS(GP)
      VOL=FN(4)
C     TEST IF BAG VOLUME LT VMIN
      P2=FDC(VOL)
C     ITERATE WITHIN TIME STEP FOR CONVERGENT SOLUTION TO RATE EQS
 1100 CONTINUE
      MCT=MCT+1
      DO 100 K=1,3
      FX(K)=FN(K)
      GX(K)=GN(K)
  100 CONTINUE
      Q(1)=P1+P2
C     WRITE(LP,66) KFLAG,P2,Q(1),TESTD
C 66  FORMAT(3X,'KFALG,P2,Q(1),TESTD:',I3,3(2X,E12.4))
C     P3: CO2 ABSORBTION
      P3=ETA*GN(2)/GN(4)*(GP-GL)
      Q(2)=-P3
      Q(3)=0.
      QQ(1)=-B1
      QQ(2)=B2
      QQ(3)=0.
C     P6: FLOW RATE FROM BAG IF BAG VOL=VMAX
      P6=FET(VOL)
      DO 450 KT=1,5
      P3=ETA*GN(2)/GN(4)*(GP-GL)
      Q(2)=-P3
C     IFLAG=1: INHALATION
C     JFLAG=1: EXHALATION
      IF(IFLAG.EQ.0) GO TO 135
      DO 125 I=1,3
      FN(I)=(F(I)+DT*Q(I))/(1.+DT/FN(4)*(FP+FL)+DT*P6/VMAX)
      GN(I)=G(I)+DT*(FN(I)/FN(4)*FP+QQ(I))
 125  CONTINUE
 135  CONTINUE
      IF(JFLAG.EQ.0) GO TO 155
      DO 150 I=1,3
      FN(I)=(F(I)+DT*(Q(I)+GN(I)/GN(4)*(GP-GL)))/(1.+DT*P6/VMAX)
      GN(I)=(G(I)+DT*QQ(I))/(1.+DT/GN(4)*GP)
 150  CONTINUE
 155  CONTINUE
C     SOLVE IMPLICIT FORM OF RATE EQUATIONS
      DO 200 I=1,3
      FN(I)=F(I)+DT*((GN(I)/GN(4)*(GP-GL)-FN(I)/FN(4)*(FP+FL))
     1      +Q(I))-DT*FN(I)*P6/VMAX
```

```
      IF(FN(I).LT.0.0) FN(I)=1.0E-20
      GN(I)=G(I)+DT*((FN(I)/FN(4)*FP-GN(I)/GN(4)*GP)+QQ(I))
      IF(GN(I).LT.0.0) GN(I)=1.0E-20
 200  CONTINUE
      SUM=0.
      SUMM=0.
      DO 400 I=1,3
      SUM=SUM+FN(I)
      SUMM=SUMM+GN(I)
 400  CONTINUE
      FN(4)=SUM
      GN(4)=SUMM
 450  CONTINUE
      IF(MCT.LT.4) GO TO 1100
      EPM=0.
C     TEST FOR COVERGENT SOLUTION
      DO 500 I=1,3
      EP(I)=ABS(FN(I)-FX(I))
      IF(EP(I).GT.EPM) EPM=EP(I)
 500  CONTINUE
      IF(EPM.GT.EPS.AND.MCT.LT.171) GO TO 1100
      IF(EPM.LT.EPS) GO TO 600
      WRITE(LP,800) NCT,MCT,EPM
 800  FORMAT(5X,'NCT=',I5,2X,'MCT=',I5,2X,'EPM=',E12.5)
      IF(MCT.LT.200) GO TO 1100
      CALL OUTPT
      STOP 'NONCONVERGENCE'
600   CONTINUE
      TIME=TIME+DT
      DO 720 I=1,4
      F(I)=FN(I)
      G(I)=GN(I)
 720  CONTINUE
C     IF(MOD(NCT,MOP).EQ.0.AND.NCT.LE.MAX) CALL OUTPT
C     IF(NCT.LE.400) CALL OUTPT
C     IF(MOD(NCT,MOP).EQ.0.AND.NCT.GE.MAX1.AND.NCT.LE.MAX2)
C    1  CALL  OUTPT
      IF(MOD(NCT,MOP).EQ.0.AND.NCT.GE.MAXL.AND.NCT.LE.MAX)
     1   CALL OUTPT
C     IF(NCT.LE.MAX) CALL OUTPT
C     IF(VOL.GT.VMAX) CALL OUTPT
      IF(ILK.EQ.1) GO TO 1210
C     IF(FN(4).GE.VMAX) CALL OUTPT
      IF(FN(4).GE.VMAX) ILK=1
      NNCT=NCT
 1210 CONTINUE
C     IF(ILK.EQ.1) CALL OUTPT
      IF(NCT.LT.MAX) GO TO 1000
      CLOSE(UNIT=ILP)
      STOP
```

```
      END
      SUBROUTINE INIT
      INCLUDE 'LUNGG.COM'
C     P1:O2 FROM BOTTLE
C     B1:O2 UPTAKE
C     B2:CO2 PRODUCTION
C     READ INPUT DATA
C     BO2: O2 PRODUCTION FROM BOTTLE,L/MIN
C     UO2: O2 UPTAKE,L/MIN
C     RQ: RESPIRATORY QUOTIENT
C     RATTE: BREATHING RATE, BPM
C     BCNO2: INITIAL O2 CONCENTRATION IN BAG
C     BCNCO2: INITIAL CO2 CONCENTRATION IN BAG
C     BCNN2: INITIAL N2 CONCENTRATION IN BAG
C     UCNO2: INITIAL O2 CONCENTRATION IN LUNG
C     UCNCO2: INITIAL CO2 CONCENTRATION IN LUNG
C     UCNN2: INITIAL N2 CONCENTRATION IN LUNG
      IO=1
      OPEN(UNIT=IO,NAME='DUMMIE.DAT',CARRIAGECONTROL='LIST',
     1     FORM='FORMATTED',TYPE='OLD',ERR=7788)
 7788 CONTINUE
      READ(IO,11) BO2,UO2,RQ,RATTE
      READ(IO,13) BCNO2,BCNCO2,BCNN2
      READ(IO,15) UCNO2,UCNCO2,UCNN2
   11 FORMAT(3(5X,F5.2),4X,F7.2)
   13 FORMAT(3(5X,F5.2))
   15 FORMAT(3(5X,F5.2))
      CLOSE(UNIT=IO)
      UCO2=UO2*RQ
      P1=BO2*1.0E+03/60.
      B1=UO2*1.0E+03/60.
      B2=UCO2*1.0E+03/60.
C     DT: TIME STEP, SEC
C     ETA: ABSORBTION COEFFICIENT
C     VMAX: BAG MAXIMUM VOLUME
C     VMIN: BAG MINIMUM VOLUME
C     VT: TIDAL VOLUME
C     FROC: LUNG VOLUME
C     FRAC: INITIAL BAG VOLUME NORMALIZED BY VMAX
C     TAUD: TIME BOTTLE DEMAND VALVE IS OPEN
C     TAUR: TIME RELIEF VALVE IS OPEN
      DT=0.1
      TAUD=0.25
      TAUR=0.25
      PI=4.*ATAN(1.)
      ETA=1.0
      MFLAG=0
      VMAX=5.E+03
      VMIN=0.2*VMAX
      VT=1.39E+03
```

```
      PDD=30.E+03/60.
      FRAC=0.9
      FROC=3.0E+03
      WRITE(LP,50) BO2,UO2,UCO2,RQ,RATTE
 50   FORMAT(5X,'O2 SUPPLY FROM BOTTLE=',E12.5,1X,'L/MIN',4X,
     1'O2 UPTAKE=',E12.5,1X,'L/MIN',/4X,'CO2 PRODUCTION=',E12.5
     2,1X,'L/MIN',4X,'RESPIRATORY QUOTIENT=',E12.5,/4X
     3,'BREATHING RATE=',E12.5,'BPM'//)
      WRITE(LP,52) TAUR,TAUD
 52   FORMAT(5X,'TAUR=',E12.5,'SEC',4X,'TAUD=',E12.5,'SEC'/)
      KFLAG=0
      RATE=RATTE/60.
      FREQ=2.*PI*RATE
      AA=0.5*VT*FREQ
      FN(1)=BCNO2*FRAC*VMAX
      FN(2)=BCNCO2*FRAC*VMAX
      FN(3)=BCNN2*FRAC*VMAX
      FN(4)=FN(1)+FN(2)+FN(3)
      GN(1)=UCNO2*FROC
      GN(2)=UCNCO2*FROC
      GN(3)=UCNN2*FROC
      GN(4)=GN(1)+GN(2)+GN(3)
      DO 100 I=1,4
      F(I)=FN(I)
      G(I)=GN(I)
 100  CONTINUE
      WRITE(LP,75)
 75   FORMAT(5X,'FN: BREATHING BAG',5X,'GN: LUNG'/5X,'1-02,2-CO2
     1,3-N2,4-TOTAL'//)
      RETURN
      END
      SUBROUTINE OUTPT
      INCLUDE 'LUNGG.COM'
C     WRITE(LP,100) TIME,NCT
 100  FORMAT(/5X,'TIME=',E12.5,'SEC',4X,'NCT=',I5)
      DO 200 K=1,4
      F4N(K)=FN(K)/FN(4)
      G4N(K)=GN(K)/GN(4)
C     WRITE(LP,300) K, FN(K),GN(K),F4N(K),G4N(K)
 300  FORMAT(5X,'K=',I4,2X,'FN(K)=',E12.5,10X,'GN(K)=',E12.5
     1,4X,'F4N(K)=',E12.5,10X,'G4N(K)=',E12.5)
 200  CONTINUE
C     WRITE(LP,909) MM,P1,P2,P3,P4,P5,P6,FP,GP
909   FORMAT(3X,'MM,P1...P6',I5,6(2X,E12.5)/5X,'FP,GP',2(3X,E12.5))
      ZIP1=F4N(1)*100.
      ZIP2=FN(4)/1000.
      WRITE(ILP,99) TIME,ZIP1,ZIP2
 99   FORMAT(3(3X,E12.5))
      RETURN
      END
```

```
      FUNCTION FDC(YY)
      INCLUDE 'LUNGG.COM'
C     DEMAND VALVE SUPPLIES O2 AT EXCESS RATE PDD IF VOL LT VMIN
C                    AND FOR TIME TAUD
      FDC=0.
      IF(KFLAG.EQ.1) GO TO 100
      KFLAG=0
      TESTD=0.
      IF(FN(4).LT.VMIN) KFLAG=1
      GO TO 200
  100 IF(KCLT.EQ.1) TESTD=TESTD+DT
      IF(TESTD.LT.TAUD) FDC=PDD
      IF(TESTD.GT.TAUD) KFLAG=0
      KCLT=0
  200 CONTINUE
      RETURN
      END
      FUNCTION FET(YY)
      INCLUDE 'LUNGG.COM'
C     RELIEF VALVE OPENS IF BAG VOLUME EQ VMAX FOR TIME TAUR
      FET=0.
      DUD=0.
      IF(YY.GT.VMAX) DUD=GP-GL-(FP+FL)
     1    +P1+P2-P3
C     WRITE(LP,99) MFLAG,YY,VMAX,DUD
C 99  FORMAT(3X,I4,3(2X,E12.5))
      DUD=ABS(DUD)
      DUDD=DUD
      IF(MFLAG.EQ.1) GO TO 100
      MFLAG=0
      TESTR=0.
      DUDD=DUD
      IF(YY.GE.VMAX)  MFLAG=1
      GO TO 200
 100  IF(MCLT.EQ.1) TESTR=TESTR+DT
      IF(TESTR.LT.TAUR) FET=DUDD
C     WRITE(LP,88) NCT,MFLAG,MCLT,TESTR,DUDD
C88   FORMAT(4X,'NCT,MFLAG,MCLT,TESTR,DUDD:',3(2X,I4),2(2X,E12.5))
      IF(TESTR.GT.TAUR) MFLAG=0
C     SET MCLT=0 SO AS NOT TO INCREASE TESTR UNLESS NEW TIME STEP
      MCLT=0
 200  CONTINUE
      RETURN
      END
```

APPENDIX C.--LISTING OF COMMON BLOCK

```
      IMPLICIT DOUBLE PRECISION (A-H,O-Z)
      COMMON/FLOW1/F(4),FN(4),G(4),GN(4),F4N(4),G4N(4),Q(3),QQ(3)
     1            ,FX(4),GX(4),EP(4)
      COMMON/FLOW2/LP,TIME,EPS,P1,P2,P3,P4,P5,P6,NCT,MCT,MAX,MOP
     1            ,B1,B2,FCOEF,GCOEF,DT,VAMX,VT,VOL,FRAC,ETA,PI
     2            ,VMAX,FREQ,VMIN,AA,FL,GL,MM,FP,GP,TAUR,TAUD
     3            ,TESTD,TESTR,MCLT,KCLT,KFLAG,MFLAG,PDD,ILP
```

☆U.S. GPO: 1985-505-019/20,002